

Degree versus Dimension for Rational Parametric Curves

Barry H. Dayton

Given a rationally parameterized curve $P(t) = (p_1(t)/\Delta(t), \dots, p_n(t)/\Delta(t))$ in \mathbb{R}^n or \mathbb{C}^n , where the p_j and Δ are polynomials, we find the dimension of the smallest linear subset of \mathbb{R}^n containing the curve. If all the p_j and Δ are of degree d or less, then it is known abstractly that this dimension is d or less and *rational normal curves* play a key role in the argument. We consider this from a computational point of view with `TransformationFunction` playing an essential part in the discussion.

■ 1. Introduction

The ancients were confused about the concepts of degree and dimension. As late as 1545 in his famous book *Ars Magna* [1], Cardano, who did not hesitate to invent imaginary numbers, in reference to his assistant Ferrari's solution of the quartic gives the following disclaimer:

Although a long series of rules might be added and a long discourse given about them, we conclude our detailed consideration with the cubic, others being merely mentioned, even if generally, in passing. For as the first power refers to a line, the square to a surface, and the cube to a solid body, it would be very foolish to go beyond this point. Nature does not permit it.

The distinction between degree and dimension was later resolved by Descartes's algebraic notation. But, in the context of parametric curves, I recently noticed a simple linear algebra proof of the following theorem:

Theorem A

Let $P(t) = (p_1(t), \dots, p_n(t))$, be a curve in \mathbb{R}^n or \mathbb{C}^n where the coordinate functions $p_j(t)$ are polynomials of degree d or less. Then for any $n > d$, the curve lies in a linear subset of \mathbb{R}^n or \mathbb{C}^n of dimension d .

This theorem, as well as many of the other facts in this article, is given in Joe Harris's book [2] from a projective geometry point of view. He also considers the degree versus dimension issue in a number of other situations. We give the linear algebra proof in Section 2.

Unfortunately, projective geometry is not computationally friendly. Instead we can view these results from an affine point of view using the built-in function `TransformationFunction` [3], which we discuss in Section 3.

We then generalize and rephrase our result in Section 4 as Theorem B. The generalization is to rational curves and we can give the dimensions of the smallest linear space containing the curve. Theorem B does clarify that, while the degree bounds the size of a linear set, the curve may lie in a smaller dimensional linear set.

In Section 5 we observe that the *rational normal curve* in \mathbb{C}^d or \mathbb{R}^d , (t^d, t^{d-1}, \dots, t) , is universal for rational curves. That is, every rational curve is a transform of a normal curve. This is very easily seen via the `TransformationFunction`. This lets us rephrase Theorem B in another useful form, where the `TransformationMatrix` can be found directly from the expression of a rational function in the form

$$Q(t) = \left(\frac{p_1(t)}{\Delta(t)}, \frac{p_2(t)}{\Delta(t)}, \dots, \frac{p_n(t)}{\Delta(t)} \right),$$

where $p_1(t), p_2(t), \dots, p_n(t)$ and the common denominator $\Delta(t)$ are all polynomials of degree d or less written in descending degree. To simplify notation we generally work with coefficients in the real numbers \mathbb{R} , but it should be understood that one could work in any subfield of the complex numbers \mathbb{C} as well. But, as immediately below, in some cases we must consider parameter values in the algebraic closure of the subfield.

Sections 5 and 6 give two applications.

The first discusses the recognition problem: given a point $q \in \mathbb{R}^n$, is $q = P(t)$ for some $t \in \mathbb{C}$? This is equivalent to the well-studied problem of finding a common solution of a family of univariate polynomials, which we do not consider here. We show that modulo the linear `TransformationFunction`, the recognition problem can often be solved in a linear space of smaller dimension.

The second example is the implicitization problem for rational functions, which is to find an implicit system that describes the ideal of the rational curve. We only sketch this, as there is no room to carefully describe the routines in [4].

In fact, this article was motivated by the author's work on implicitization of parametric curves. I noticed that an unexpectedly large number of linear equations appeared in the implicit systems.

■ 2. Special Case of Polynomial Parameters

In this article a *linear subset* of \mathbb{R}^n is a set defined by a system of linear equations, not necessarily homogeneous. A linear subset is distinguished from a *linear subspace*, which is a subspace of the vector space \mathbb{R}^n and defined with homogeneous equations. The big difference is that a subspace contains the origin $(0, \dots, 0)$. A linear subset is a coset of a linear subspace under the operation of vector addition.

A polynomial parametric curve is a function $P(t) = (p_1(t), \dots, p_n(t))$ where each coordinate function is a polynomial that we write in descending degree:

$$p_j(t) = a_{j,1} t^{d_j} + a_{j,2} t^{d_j-1} + \dots + a_{j,d_j} t^1 + a_{j,d_j+1},$$

where d_j is the *degree* of the coordinate polynomial. The largest such degree is the degree of the parameterization. Note that $P(0) = (a_{1,d+1}, a_{2,d+1}, \dots, a_{j,d+1})$. This constant acts merely as a basepoint; a different basepoint gives a curve that is a translation of the first. Thus the basepoint does not affect the geometry. We say our parameterization is *stripped* if $P(0) = 0$ (or alternatively if each $a_{j,d+1} = 0$). Each polynomial parameterized curve is then a translate of a stripped curve, so we first consider those. We *strip* a polynomial parameterized curve by dropping all the constant terms.

We now create a stripped coefficient matrix M from the stripped polynomial. If d is the degree of the polynomial, M is the $n \times d$ matrix with rows $(a_{j,1}, \dots, a_{j,d})$. Consider the following equation where points are column vectors.

$$p(t) = \begin{pmatrix} t^d a_{11} + t^{d-1} a_{12} + \dots + t a_{1d} \\ t^d a_{21} + t^{d-1} a_{22} + \dots + t a_{2d} \\ t^d a_{31} + t^{d-1} a_{32} + \dots + t a_{3d} \\ \vdots \\ t^d a_{n1} + t^{d-1} a_{n2} + \dots + t a_{nd} \end{pmatrix} =$$

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1d} \\ a_{21} & a_{22} & \dots & a_{2d} \\ a_{31} & a_{32} & \dots & a_{3d} \\ \vdots & \vdots & \dots & \vdots \\ a_{n1} & a_{n2} & \dots & a_{nd} \end{pmatrix} \cdot \begin{pmatrix} t^d \\ t^{d-1} \\ t^{d-2} \\ \vdots \\ t \end{pmatrix} = t^d \begin{pmatrix} a_{11} \\ a_{21} \\ a_{31} \\ \vdots \\ a_{n1} \end{pmatrix} + t^{d-1} \begin{pmatrix} a_{12} \\ a_{22} \\ a_{32} \\ \vdots \\ a_{n2} \end{pmatrix} + \dots + t \begin{pmatrix} a_{1d} \\ a_{2d} \\ a_{3d} \\ \vdots \\ a_{nd} \end{pmatrix}$$

This shows that every point on the parameterized curve is in the vector space spanned by the columns of the coefficient matrix. So Theorem A is true for a stripped parameterization, but adding back the constant $P(0)$ simply moves this subspace to a linear subset.

To describe the smallest linear set containing a finite set of points in terms of a system of equations, here is a short routine.

```
In[1]:= linearSet[P_, X_] := Module[
  {n, r, M, U, S, V, rk, eq0},
  {n, r} = {Length[X], Length[P]};
  M = Table[
    Switch[
      Length[P[[i]]],
      n, Append[P[[i]], 1],
      n + 1, P[[i]]
    ], {i, r}];
  {U, S, V} = SingularValueDecomposition@N@M;
  rk = Length@Select[Diagonal@S, # > 1.*^-10 &];
  If[
    rk > n,
    {},
    eq0 = Take[Transpose@V, rk - n - 1, n].X;
    eq0 - (eq0 /. Thread[X → P[[1]])]
  ]
]
```

A longer version of this with error detecting is `linearSetMD` in [4].

Example 1:

```
In[2]:= P1 = {2 t + 5, 3 t^2 - 7, 3 t - 6 t^2 - 1};
```

We know a linear set containing this curve must be of dimension no greater than three, since this set is contained in \mathbb{R}^3 , so it is generated as a linear set by four or fewer points. Therefore it is enough to take four random points on this curve and calculate the smallest linear set containing them.

Here are the four random points.

```
In[3]:= pts = Table[P1 /. {t → RandomReal[{-5, 5}]}, {4}]
```

```
Out[3]= {{3.31739, -4.87661, -7.7707},
  {13.026, 41.3121, -85.5852}, {3.93328, -6.14658, -4.30692},
  {7.29616, -3.04575, -5.46426}}
```

Here is the linear expression for the linear set.

```
In[4]:= First@linearSet[pts, {x, y, z}]
```

```
Out[4]= 0.992916 - 0.0661944 x + 0.0882592 y + 0.0441296 z
```

A linear set defined by one linear equation in three variables is of dimension two. This curve lies in the linear set defined by setting the linear expression to zero.

■ 3. Rational Parametric Curves via TransformationFunction

The central concept in this article is the built-in Wolfram Language function `TransformationFunction`. When we say *transformation function* we mean a function given by `TransformationFunction`. Basically these are affine versions of *projective linear* transformations, which can include translations along with the usual transformations of linear algebra. They appeared in Lecture 2 of Abhyankar [5] and much of the author's work [4, 6] as *fractional linear transformations*; they are also known in the literature as *linear fractional transformations*. Our major use of these transformations is to be able to access projective geometry where points are cosets of $(n + 1)$ -tuples, while working in affine geometry where points are merely n -tuples, which are easy to manipulate computationally.

A transformation function $\mathbb{R}^m \rightarrow \mathbb{R}^n$ can be described by an $(n + 1) \times (m + 1)$ matrix. The matrix of the associated projective linear transformation is called the *transformation matrix* in the Wolfram Language. Thus the `TransformationFunction` of an $(n + 1) \times (m + 1)$ matrix A takes an affine m -tuple, appends 1 to represent this in projective m -space, applies the projective linear transformation defined by A and then specializes by dividing by the $(n + 1)^{\text{st}}$ component.

Here is an example.

$$\text{In[5]:= TransformationFunction}\left[\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ -1 & -2 & -3 \end{pmatrix}\right][\{\mathbf{x}, \mathbf{y}\}]$$

$$\text{Out[5]= } \left\{ \frac{3 + \mathbf{x} + 2 \mathbf{y}}{-3 - \mathbf{x} - 2 \mathbf{y}}, \frac{6 + 4 \mathbf{x} + 5 \mathbf{y}}{-3 - \mathbf{x} - 2 \mathbf{y}} \right\}$$

These transformations in the special case $\mathbb{R}^2 \rightarrow \mathbb{R}^2$ are discussed in detail in Chapter 6 of my book [4].

A transformation function is *affine* if the last row is $(0, \dots, 0, 1)$; the denominators are always 1, the upper-left $n \times m$ submatrix gives a linear transformation and the first n entries of the last column describe a translation.

$$\text{In[6]:= TransformationFunction}\left[\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 0 & 0 & 1 \end{pmatrix}\right][\{\mathbf{x}, \mathbf{y}\}]$$

$$\text{Out[6]= } \{3 + \mathbf{x} + 2 \mathbf{y}, 6 + 4 \mathbf{x} + 5 \mathbf{y}\}$$

In particular, the domain of an affine transformation $\mathbb{R}^m \rightarrow \mathbb{R}^n$ is all of \mathbb{R}^m . Otherwise we call the transformation function *projective*. If the last row of the transformation matrix is $(a_{n+1,1}, \dots, a_{n+1,m+1})$, then the hyperplane of \mathbb{R}^m given by $a_{n+1,1}x_1 + \dots + a_{n+1,m}x_m + a_{n+1,m+1} = 0$ is not in the domain of the transformation function. In the context of an affine transformation, it is understood that the equation $1 = 0$ defines the empty set.

In this article we assume that a rational parametric curve has coordinates that are quotients of two polynomials in t . We insist that the parametric curve be given with a common denominator $\Delta(t)$, so, for example, is of the form

$$Q(t) = \left(\frac{p_1(t)}{\Delta(t)}, \frac{p_2(t)}{\Delta(t)}, \dots, \frac{p_n(t)}{\Delta(t)} \right) \quad (1)$$

for polynomials p_1, \dots, p_n, Δ . The degrees of $p_j(t)$ may be greater than, equal to or less than the degree of $\Delta(t)$. In particular, $\Delta(t)$ could be the constant polynomial 1, in which case $Q(t)$ is a polynomial curve that we can treat as a special case of a rational curve. The degree of $Q(t)$ is the largest degree of p_1, \dots, p_n, Δ .

The advantage of writing polynomials in the parameter t in descending degree is that writing a transformation matrix for a rational function is easy. Suppose in equation (1) that $p_j = a_{j,1}t^d + \dots + a_{j,d}t + a_{j,d+1}$ for $j = 1, \dots, d+1$, where we write $\Delta = p_{n+1}$. Then the transformation matrix for $Q[t]$ is

$$\begin{pmatrix} a_{11} & a_{12} & \cdots & a_{1,d} & a_{1,1+d} \\ a_{21} & a_{22} & \cdots & a_{2,d} & a_{2,1+d} \\ \vdots & \vdots & \cdots & \vdots & \vdots \\ a_{1+n,1} & a_{1+n,2} & \cdots & a_{1+n,d} & a_{1+n,1+d} \end{pmatrix}. \quad (2)$$

Example 2.

$$\text{In}[7]:= \mathbf{Q} = \left\{ \frac{t^3 - 2t^2 + 4t - 3}{t^2 + 1}, \frac{4t^2 - 4}{t^2 + 1}, \frac{-t^2 + 2}{t^2 + 1}, \frac{8}{t^2 + 1} \right\};$$

Here is the transformation matrix.

```
In[8]:= (QA = MatrixForm[{{1, -2, 4, -3}, {0, 4, 0, -4},
{0, -1, 0, 2}, {0, 0, 0, 8}, {0, 1, 0, 1}}]) //
MatrixForm
```

Out[8]/MatrixForm=

$$\begin{pmatrix} 1 & -2 & 4 & -3 \\ 0 & 4 & 0 & -4 \\ 0 & -1 & 0 & 2 \\ 0 & 0 & 0 & 8 \\ 0 & 1 & 0 & 1 \end{pmatrix}$$

This is the curve.

```
In[9]:= TransformationFunction[QA] [{t3, t2, t}]

Out[9]:= TransformationFunction[ $\begin{pmatrix} 1 & -2 & 4 & -3 \\ 0 & 4 & 0 & -4 \\ 0 & -1 & 0 & 2 \\ 0 & 0 & 0 & 8 \\ 0 & 1 & 0 & 1 \end{pmatrix}$ ] [{t3, t2, t}]
```

Both [2] and [5] mention the fact that every rationally parameterized curve is a projective transformation applied to a polynomially parameterized curve. In particular, [2] notes that this polynomial curve can be the rational normal curve of degree d .

■ 4. Theorem B

Before we state Theorem B, we note that every linear transformation can be factored into a projection on some coordinates followed by an embedding. This is accomplished in a special way using Mathematica by the following matrix reduction algorithm we call `rankDecomposition`. This takes an $n \times d$ matrix A of rank r and outputs an $n \times r$ matrix L and an $r \times d$ matrix R consisting of rows of A such that $LR = A$. This implies that r rows of A are what the Wolfram Language calls `UnitVectors`; that is, L contains an $r \times r$ identity matrix as a submatrix.

In the code, the functions `pivotrows` and `nonpivotrows` defined in the `Do` statements invert the lists `pivots` and `nonpivots` viewed as functions from their index sets. The `IntegerQ` tests whether k is in the domain of `pivotrows`.

```
In[10]:= rankDecomposition[A_] := Module[
  {n, prenulls, pivots, nonpivots, pivotrows, nonpivotrows,
   R, L},
  n = First@Dimensions@A;
  If[MatrixRank@A == n, Return[{IdentityMatrix[n], A}];
  prenulls =
    Map[Reverse, RowReduce@NullSpace@Transpose@Reverse@A];
  pivots = Flatten@Map[FirstPosition[#, 1, 1] &, prenulls];
  nonpivots = Complement[Range@n, pivots];
  Do[pivotrows@pivots[[i]] = i, {i, Length@pivots}];
  Do[nonpivotrows@nonpivots[[j]] = j,
    {j, Length@nonpivots}];
  R = A[[nonpivots]];
  L = Table[
    If[
      IntegerQ[pivotrows@k],
      LinearSolve[Transpose@R, A[[k]]],
      UnitVector[Length@nonpivots, nonpivotrows@k]
    ],
    {k, n}];
  {L, R}
]
```

We can now state and prove our main theorem; we write $p_{n+1}(t) = \Delta(t)$. It may seem counterintuitive that we can strip the constant off the denominator, in particular for polynomially parameterized curves $\Delta(t) = 1$ (so stripping it gives $\tilde{\Delta} = 0$). But projectively the denominator is just another coordinate so we can still do that. So if A is the matrix from the previous section and $P(t) = F((t^d, \dots, t))$, where F is the `TransformationFunction` of A , then the *projective stripped coefficient matrix* of P is just the submatrix of A with the last column removed.


Theorem B

Let $P(t) = (p_1(t)/\Delta(t), \dots, p_n(t)/\Delta(t))$ be a parametric curve in \mathbb{R}^n of degree d . Suppose the projective stripped coefficient matrix of P has rank r . Then there are r components of P defining a stripped polynomial parametric curve $\rho(t) = (\tilde{p}_{i_1}, \tilde{p}_{i_2}, \dots, \tilde{p}_{i_r})$ in \mathbb{R}^r and a transformation function $F : \mathbb{R}^r \rightarrow \mathbb{R}^n$ taking $\rho(t)$ to $P(t)$.

Proof

We apply the algorithm `rankDecomposition` to the $(n+1) \times d$ projective stripped coefficient matrix M of P , obtaining a list of r rows forming a basis R_M of the row space of M and a matrix L_M of size $(n+1) \times r$, where the rows corresponding to this basis are replaced by rows of the $r \times r$ identity matrix. Multiplying R_M by the vector (t^d, t^{d-1}, \dots, t) gives the parametric function $p(t)$. Appending a last column to L_M with the constant terms of the original $P(t)$ gives a transformation matrix L_A . By the above comments it is easy to see that the `TransformationFunction` F defined by L_A takes $\rho(t)$ to $P(t)$.

One can paraphrase this theorem as: *Given a parametric curve $P(t)$ of degree d , there is an $r \leq d$, a stripped parametric polynomial curve ρ in \mathbb{R}^r and a `TransformationFunction` F so that the following diagram commutes.*



We ask the reader not to take this diagram literally in the case of a rational parameterization, as the domains of ρ , F , P may not be the full spaces indicated. But if P is a polynomial parameterization, then the domains are the full spaces and F is an embedding.

Example 3: We illustrate this proof by fully working out the following degree-two curve in \mathbb{R}^3 .

$$\text{In}[11]:= \mathbf{P1} = \{2t + 5, 3t^2 - 7, 3t - 6t^2 - 1\};$$

$$\text{In}[12]:= \mathbf{M1} = \begin{pmatrix} 2 & 0 \\ 0 & 3 \\ 3 & -6 \\ 0 & 0 \end{pmatrix};$$

The decomposition can be easily done by hand.

$$\text{In}[13]:= \text{MatrixForm} /@ (\{\mathbf{LM1}, \mathbf{RM1}\} = \text{rankDecomposition}[\mathbf{M1}])$$

$$\text{Out}[13]= \left\{ \begin{pmatrix} 1 & 0 \\ 0 & 1 \\ \frac{3}{2} & -2 \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 2 & 0 \\ 0 & 3 \end{pmatrix} \right\}$$

So we add the constant row; remember that the constant in the last row is 1.

$$\begin{aligned} \text{In}[14]:= \rho1 &= \{2t, 3t^2\}; \\ (\mathbf{LA1} &= \text{Join}[\mathbf{LM1}, \{\{5\}, \{-7\}, \{-1\}, \{1\}\}, 2]) // \\ &\text{MatrixForm} \end{aligned}$$

Out[15]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 5 \\ 0 & 1 & -7 \\ \frac{3}{2} & -2 & -1 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\text{In}[16]:= \mathbf{F1} = \text{TransformationFunction}[\mathbf{LA1}]$$

$$\text{Out}[16]= \text{TransformationFunction} \left[\left(\begin{array}{cc|c} 1 & 0 & 5 \\ 0 & 1 & -7 \\ \frac{3}{2} & -2 & -1 \\ \hline 0 & 0 & 1 \end{array} \right) \right]$$

Theorem B tells us the composition of $\mathbf{F1}$ and $\rho1$ is $\mathbf{P1}$.

$$\text{In}[17]:= \mathbf{F1}[\rho1]$$


$$\text{Out}[17]= \{5 + 2t, -7 + 3t^2, -1 + 3t - 6t^2\}$$

```
In[18]:= form1 = linearSet[F1 /@ {{0, 0}, {1, 0}, {0, 1}}, {x, y, z}]
```

```
Out[18]:= {0.992916 - 0.0661944 x + 0.0882592 y + 0.0441296 z}
```

So this curve is contained in a plane.

```
In[19]:= Show[
  ContourPlot3D[form1 == 0, {x, 3, 8}, {y, -7, 5},
    {z, -10, 0}, Mesh -> None, ContourStyle -> Opacity[.2]],
  ParametricPlot3D[P1, {t, -1, 2}, PlotStyle -> Thick]
]
```



Example 2 (continued): We now consider the rational parameterization Q of example 2.

```
In[20]:= Q = { $\frac{-3 + 4t - 2t^2 + t^3}{1 + t^2}$ ,  $\frac{-4 + 4t^2}{1 + t^2}$ ,  $\frac{2 - t^2}{1 + t^2}$ ,  $\frac{8}{1 + t^2}$ };
```

```
In[21]:= (QA = {{1, -2, 4, -3}, {0, 4, 0, -4}, {0, -1, 0, 2},
  {0, 0, 0, 8}, {0, 1, 0, 1}}) // MatrixForm
```

```
Out[21]//MatrixForm=
```

$$\begin{pmatrix} 1 & -2 & 4 & -3 \\ 0 & 4 & 0 & -4 \\ 0 & -1 & 0 & 2 \\ 0 & 0 & 0 & 8 \\ 0 & 1 & 0 & 1 \end{pmatrix}$$

```
In[22]:= QM = Take[QA, All, 3]
```

```
Out[22]= {{1, -2, 4}, {0, 4, 0}, {0, -1, 0}, {0, 0, 0}, {0, 1, 0}}
```

```
In[23]:= QA0 = Take[QA, All, -1]
```

```
Out[23]= {{-3}, {-4}, {2}, {8}, {1}}
```

```
In[24]:= {LM, RM} = rankDecomposition[QM]
```

```
Out[24]= {{{{1, 0}, {0, 1}, {0, -1/4}, {0, 0}, {0, 1/4}},
           {{1, -2, 4}, {0, 4, 0}}}
```

```
In[25]:= LA = Join[LM, QA0, 2]
```

```
Out[25]= {{1, 0, -3}, {0, 1, -4}, {0, -1/4, 2}, {0, 0, 8}, {0, 1/4, 1}}
```

```
In[26]:= Qp = RM.{t^3, t^2, t}
```

```
Out[26]= {4 t - 2 t^2 + t^3, 4 t^2}
```

```
In[27]:= TransformationFunction[LA][Qp]
```

```
Out[27]= { $\frac{-3 + 4 t - 2 t^2 + t^3}{1 + t^2}$ ,  $\frac{-4 + 4 t^2}{1 + t^2}$ ,  $\frac{2 - t^2}{1 + t^2}$ ,  $\frac{8}{1 + t^2}$ }
```

We check that this lies in a two-dimensional plane in \mathbb{R}^4 .

```
In[28]:= QPts = Table[Q /. {t -> p}, {p, RandomReal[{-6, 6}, 5]}]
```

```
Out[28]= {{{-5.7033, 2.9341, -0.600287, 1.0659},
           {-7.88708, 3.72568, -0.89713, 0.274321},
           {-8.30167, 3.76652, -0.912445, 0.23348},
           {3.26564, 3.65326, -0.869973, 0.346738},
           {-6.97704, 3.58161, -0.843102, 0.418394}}
```

```
In[29]:= linearSet[QPts, {x, y, z, w}] // Chop
```

```
Out[29]= {-0.821813 + 0.149354 w + 0.317651 y + 0.448792 z,
           0.462848 - 0.222283 w + 0.0974295 y + 0.852566 z}
```

The step in the proof of Theorem B where we use R_M to obtain the curve ρ from the rational normal curve can also be done by using an affine transformation function obtained by adding the row and column $(0, \dots, 0, 1)$ to R_M . In Example 2 we have the following.

```
In[30]:= RA = Join[Append[RM, {0, 0, 0}], {{0}, {0}, {1}}, 2]
```

```
Out[30]= {{1, -2, 4, 0}, {0, 4, 0, 0}, {0, 0, 0, 1}}
```


```
In[31]:= TransformationFunction[RA][{t^3, t^2, t}]
```

```
Out[31]= {4 t - 2 t^2 + t^3, 4 t^2}
```

This gives:

Theorem C

Let $P: \mathbb{R} \rightarrow \mathbb{R}^n$ be a rational (or polynomial) curve parameterization of degree d . Suppose the projective stripped coefficient matrix of $P(t)$ has rank r . Then the transformation function F in Theorem B can be decomposed into transformation functions $F_2 \circ F_1$ as in the following diagram.



Here F_1 is an affine transformation function of \mathbb{R}^d onto \mathbb{R}^r and F_2 is a possibly projective transformation of \mathbb{R}^r into \mathbb{R}^n . In particular, the parametric curve given by P lies in a linear subset of \mathbb{R}^n of dimension less than or equal to the minimum of n, r, d .

Construction

As in Theorem B, we let M be the projective stripped matrix of $P(t)$ and apply rankDecomposition to M to get (L_M, R_M) of sizes $(n+1) \times r$ and $r \times d$, respectively. Appending a row of zeros and then a column of zeros with last component 1 to make R_M into an affine transformation matrix R_{MA} of size $(r+1) \times (d+1)$, let F_1 be the TransformationMatrix of R_{MA} . Appending the column of constants to L_M , we get a transformation matrix $L_{M,T}$ of size $(n+1) \times (r+1)$. Then F_2 is the TransformationMatrix of $L_{M,T}$. One can check that $P(t) = F_2(F_1((t^d, t^{d-1}, \dots, t)))$.

This recovers the known result [2] that every rational parameterization is a projective linear transformation of the rational normal curve, but here we have a constructive approach.

Example 4: For an easy but nontrivial (i.e. not conic) example we use the *piriform* [7].

$$\text{In[32]:= } \mathbf{P3} = \left\{ \frac{1 - t^4}{1 + 2t^2 + t^4}, \frac{4t}{1 + 2t^2 + t^4} \right\};$$

Here $d = 4$, $n = 2$. Here is the stripped projective matrix.

$$\text{In[33]:= } \mathbf{M3} = \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 4 \\ 1 & 0 & 2 & 0 \end{pmatrix};$$

A trivial application of rankDecomposition in that M3 is of full rank gives the following.

`In[34]:= MatrixForm /@ ({LM3, RM3} = rankDecomposition[M3])`


$$\text{Out[34]= } \left\{ \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 4 \\ 1 & 0 & 2 & 0 \end{pmatrix} \right\}$$

`In[35]:= (ARM3 = Join[Append[RM3, {0, 0, 0, 0}], {{0}, {0}, {0}, {1}}, 2]) // MatrixForm`

`Out[35]//MatrixForm=`

$$\begin{pmatrix} -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 4 & 0 \\ 1 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

`In[36]:= F1 = TransformationFunction[ARM3]`

`Out[36]= TransformationFunction[ Dimensions: 4 → 3
Type: Affine]`

`In[37]:= (TLM3 = Join[LM3, {{1}, {0}, {1}}, 2]) // MatrixForm`

`Out[37]//MatrixForm=`

$$\begin{pmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 1 \end{pmatrix}$$

```
In[38]:= F2 = TransformationFunction[TLM3]
```

```
Out[38]= TransformationFunction[ $\left[ \begin{array}{ccc|c} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 1 \end{array} \right]$ ]
```

```
In[39]:= F2@F1@{t^4, t^3, t^2, t}
```

```
Out[39]=  $\left\{ \frac{1 - t^4}{1 + 2 t^2 + t^4}, \frac{4 t}{1 + 2 t^2 + t^4} \right\}$ 
```

Notice here that $n = 2$, $r = 3$ and $d = 4$. In this case, the curve lies in \mathbb{R}^2 , a two-dimensional space. The numbers n , r , d are important values in describing a rational parameterized curve. Even though the transformation matrix for F_2 contains the 3×3 identity matrix, it is not injective, which is typical in the case of a rational parameterization, even when $n \geq r$, but this does not occur for a polynomial parameterization.

■ 5. The Recognition Problem

The recognition problem is: *given a parameterized curve $P : \mathbb{R} \rightarrow \mathbb{R}^n$ and a point q in \mathbb{R}^n , is q in the curve*; that is, does there exist t with $P(t) = q$?

There are two obvious methods to solve this problem. The first is to directly solve the over-determined system $P(t) = q$ using `NSolve[P - q]`. This works surprisingly well, failing mostly with poorly conditioned systems for which the other methods following may not work well either. The biggest problem with this approach is that when it does not work, it gives a false negative to the recognition problem. One can, of course, solve component by component and see if any solutions are numerically close.

Example 2 continued.

```
In[40]:= Q =  $\left\{ \frac{t^3 - 2 t^2 + 4 t - 3}{t^2 + 1}, \frac{4 t^2 - 4}{t^2 + 1}, \frac{-t^2 + 2}{t^2 + 1}, \frac{8}{t^2 + 1} \right\};$ 
q1 = {1.2616272901771202`, 2.75550189053762`,
      -0.5333132089516074`, 1.2444981094623802`}
q2 = {0.9328657844018655`, 3.6099079667742315`,
      -0.0055749054767041395`, 0.3132677134420798`}
```

```
Out[41]= {1.26163, 2.7555, -0.533313, 1.2445}
```

```
Out[42]= {0.932866, 3.60991, -0.00557491, 0.313268}
```

In[43]:= **NSolve**[**Q** - **q1**]

Out[43]:= {{**t** → 2.32987}}

In[44]:= **NSolve**[**Q** - **q2**]

Out[44]:= {}

So the first point $P[q1]$ is on the curve but the second point $P[q2]$ is not. In general, finding a common zero of a set of polynomial or rational equations is an interesting problem, but we do not consider that here.

The second method is to find a system of equations whose solution set is the Zariski closure of the point set $\{P(t) : t \in \mathbb{R}\}$. All that then needs to be done, in principle, is to evaluate this system at q and check that the value is 0. We consider this issue in Section 5.

As we have seen, a parameterized curve in \mathbb{R}^n may lie in a linear subset of dimension less than n . Using Theorem C and the `linearSet` algorithm, we can get some additional information about the problem and perhaps reduce this to a problem in a smaller \mathbb{R}^r .

Example 5.

In[45]:= **P4** = $\left\{ \frac{2 - 3t + 2t^2 + t^3}{1 + t^2 + 2t^3}, \frac{-4 + t^2 - 2t^3}{1 + t^2 + 2t^3}, \frac{-2 - 3t + t^2 - t^3}{1 + t^2 + 2t^3}, \frac{-1 + t - t^2 + t^3}{1 + t^2 + 2t^3} \right\};$

We would like to find out which, if any, of the following points are on this curve.

In[46]:= **q41** = $\left\{ \frac{4}{7}, -\frac{16}{21}, -\frac{4}{7}, \frac{5}{21} \right\};$
q42 = $\left\{ \frac{2}{5}, -\frac{14}{15}, -\frac{4}{5}, \frac{1}{3} \right\};$
q43 = $\left\{ \frac{3}{7}, \frac{4}{3}, -\frac{2}{5}, 1 \right\};$

We first find the transformation functions. Here is the projective stripped coefficient matrix.

In[49]:= **M4** = $\begin{pmatrix} 1 & 2 & -3 \\ -2 & 1 & 0 \\ -1 & 1 & -3 \\ 1 & -1 & 1 \\ 2 & 1 & 0 \end{pmatrix};$

In[50]:= **con4** = {{**2**}, {-**4**}, {-**2**}, {-**1**}, {**1**}};

Apply rankDecomposition.

```
In[51]:= MatrixForm /@ ({L4, R4} = rankDecomposition[M4])
```

$$\text{Out[51]} = \left\{ \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -\frac{1}{3} & -\frac{1}{2} & \frac{1}{6} \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 1 & 2 & -3 \\ -2 & 1 & 0 \\ 2 & 1 & 0 \end{pmatrix} \right\}$$

Augment these matrices to get transformation matrices.

```
In[52]:= (R4a = Join[L4, con4, 2]) // MatrixForm
```

```
Out[52]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & -4 \\ 1 & 0 & -1 & -2 \\ -\frac{1}{3} & -\frac{1}{2} & \frac{1}{6} & -1 \\ 0 & 0 & 1 & 1 \end{pmatrix}$$

```
In[53]:= (L4a = Join[Append[R4, {0, 0, 0}], {{0}, {0}, {0}, {1}}, 2]) // MatrixForm
```

```
Out[53]//MatrixForm=
```

$$\begin{pmatrix} 1 & 2 & -3 & 0 \\ -2 & 1 & 0 & 0 \\ 2 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
In[54]:=
```

```
F41 = TransformationFunction[L4a]
```

$$\text{Out[54]} = \text{TransformationFunction} \left[\left(\begin{array}{ccc|c} 1 & 2 & -3 & 0 \\ -2 & 1 & 0 & 0 \\ 2 & 1 & 0 & 0 \\ \hline 0 & 0 & 0 & 1 \end{array} \right) \right]$$

```
In[55]:= F42 = TransformationFunction[R4a]
```

$$\text{Out[55]} = \text{TransformationFunction} \left[\left(\begin{array}{ccc|c} \text{[Matrix]} & & & \end{array} \right) \right] \quad \begin{array}{l} \text{Dimensions: } 3 \mapsto 4 \\ \text{Type: LinearFractional} \end{array}$$

```
In[56]:= Simplify[F42@F41@{t^3, t^2, t}]
```

$$\text{Out[56]} = \left\{ \frac{2 - 3t + 2t^2 + t^3}{1 + t^2 + 2t^3}, \frac{-4 + t^2 - 2t^3}{1 + t^2 + 2t^3}, -\frac{2 + 3t - t^2 + t^3}{1 + t^2 + 2t^3}, \frac{-1 + t - t^2 + t^3}{1 + t^2 + 2t^3} \right\}$$

Generate some random points.

```
In[57]:= pts3 = RandomReal[{-1, 1}, {4, 3}];
```

```
In[58]:= lin41 = linearSet[F41/@pts3, {x, y, z}]
```

```
Out[58]= {}
```

```
In[59]:= lin42 = linearSet[F42/@pts3, {x, y, z, w}]
```

```
Out[59]= {0.481932 - 0.481932 w - 0.562254 x - 0.240966 y + 0.40161 z}
```

This says F41 is not contained in any proper subspace, but the image of F42 lies in a three-dimensional subspace of \mathbb{R}^4 .

```
In[60]:= lin42 /. Thread[{x, y, z, w} → q41]
```

```
Out[60]= {-3.05311 × 10-16}
```

```
In[61]:= lin42 /. Thread[{x, y, z, w} → q42]
```

```
Out[61]= {-1.66533 × 10-16}
```

```
In[62]:= lin42 /. Thread[{x, y, z, w} → q43]
```

```
Out[62]= {-0.722897}
```

The points q41 and q42 do lie in the image of F42[F41], so may be points on the curve, but we can eliminate q43. We find the fibers (preimage) of q41 and q42 in \mathbb{R}^3 .

```
In[63]:= form42 = Simplify@F42@{x1, x2, x3}
```

```
Out[63]= { $\frac{2 + x1}{1 + x3}$ ,  $\frac{-4 + x2}{1 + x3}$ ,  $\frac{-2 + x1 - x3}{1 + x3}$ ,  $\frac{-6 - 2 x1 - 3 x2 + x3}{6 (1 + x3)}$ }
```

```
In[64]:= sol1 = {x1, x2, x3} /. First@Solve[form42 == q41]
```

```
Out[64]= {10, -12, 20}
```

```
In[65]:= sol2 = {x1, x2, x3} /. First@Solve[form42 == q42]
```

```
Out[65]= {4, -10, 14}
```

These conveniently are singleton points. Thus we have reduced this rational recognition problem in \mathbb{R}^4 to a polynomial problem recognition problem in \mathbb{R}^3 .

```
In[66]:= ρ = F41@{t^3, t^2, t}
```

```
Out[66]= {-3 t + 2 t^2 + t^3, t^2 - 2 t^3, t^2 + 2 t^3}
```

```
In[67]:= NSolve[ρ - sol1]
```

```
Out[67]= {{t -> 2.}}
```

```
In[68]:= NSolve[ρ - sol2]
```

```
Out[68]= {}
```

So $q41 = P4[2]$ but $q42$ is not on the curve.

■ 6. Implicitization of Rational Parametric Space Curves

As mentioned, the motivation for this article is my work on implicitization of rational parametric space curves. In this section I only sketch my algorithms; details are in [4]. The key here is that by the material discussed, especially Theorem C, every such curve is simply a fractional linear transformation of the rational normal curve.

By *implicitization* I mean describing these parametric curves by way of algebraic equations. A problem that arises is that while one expects a curve in \mathbb{R}^n to be given by $n - 1$ equations, this is often not enough to fully describe the curve pointwise or algebraically. The standard counterexample is the *twisted cubic*, which is just the rational normal curve of degree three, $T_3(t) = (t^3, t^2, t)$. A system of three equations in the variables x, y, z describing the twisted cubic, given in [2], is $(xz - y^2, x - yz, y - z^2)$. An exercise in [2] is to show that the zero set of any pair of these three equations contains not only the twisted cubic, but also a line, but note that the extra line in the last pair lies in the infinite plane of projective three-space.

Any implicitization problem has infinitely many possible answers, but the best answers are systems of equations that form an H-basis. This idea goes back to F. S. Macaulay in 1916, who was studying homogeneous equations, hence the ‘‘H’’; basically in our context it means that any equation f of total degree d containing the parametric curve in its zero set is a polynomial combination $f = g_1 h_1 + \cdots + g_k h_k$, where the h_i are in the H-basis and the g_i are polynomials so that each term $g_i h_i$ has total degree at most d . Thus for an H-basis, the ideal membership problem reduces to linear algebra.

If one has a system S with zero set describing the parametric curve $P(t)$, then the Gröbner basis with respect to a degree ordering is an H-basis, perhaps larger than necessary. In practical terms one can simply use the following format.

```
In[69]:= H = GroebnerBasis[S, MonomialOrder → DegreeLexicographic];
```

In the case of the rational normal curve of degree d , Harris [2] claims that using quadratic equations is sufficient, so we can proceed as follows: we first give a procedure finding the total degree d of a polynomial of several variables.

```
In[70]:= totalDegree[f_, X_] :=  
Max[Total /@ Keys@CoefficientRules[f, X]]
```

Then we use the following code, say for $d = 3$.

```
In[71]:= With[{d = 3},  
S = Table[x[i] - x[d]^(d + 1 - i), {i, d - 1}]]
```

```
Out[71]= {x[1] - x[3]^3, x[2] - x[3]^2}
```

This defines H .

```
In[72]:= With[{d = 3},  
Select[  
  GroebnerBasis[S, MonomialOrder → DegreeLexicographic],  
  totalDegree[#, Table[x[i], {i, d}]] == 2 &  
]  
]
```

```
Out[72]= {-x[1] + x[2] × x[3], -x[2] + x[3]^2, -x[2]^2 + x[1] × x[3]}
```

Likewise we get the following for $d = 4, 5$.

```
In[73]:= {-x[1] + x[3]^2, -x[2] + x[3] × x[4], -x[1] + x[2] × x[4],  
-x[3] + x[4]^2, -x[2]^2 + x[1] × x[3], -x[2] × x[3] + x[1] × x[4]};
```

```
In[74]:= {-x[2] + x[4]^2, -x[1] + x[3] × x[4], -x[3]^2 + x[2] × x[4],  
-x[3] + x[4] × x[5], -x[2] + x[3] × x[5], -x[1] + x[2] × x[5],  
-x[4] + x[5]^2, -x[2] × x[3] + x[1] × x[4], -x[2]^2 + x[1] × x[3],  
-x[3]^2 + x[1] × x[5]};
```

The size of the H-basis H_d is $\frac{d(d-1)}{2}$, which gets much larger than $d-1$. The numbers $\frac{d(d-1)}{2}$ are binomial coefficients and can be enumerated recursively; however, H_{d+1} does not contain H_d , so there is no obvious recursive construction of these bases.

In [4] I construct, for the fractional linear transformation given by an $(n + 1) \times (m + 1)$ transformation matrix A , a transformation `FLTMD[F, A, k, X, Y, tol]`. This takes the system F in \mathbb{R}^m with variables given by the list X to a system in \mathbb{R}^n with variable list Y such that for a solution \mathbf{x}_0 of F then `TransformationFunction[A][x0]` is a solution of G . Unfortunately, this works numerically and the user must provide a number k that bounds the degrees of the polynomials used and a small tolerance `tol`, but for an appropriate choice of these parameters the system G is often an H-basis if F is.

Thus, a possible method for finding an implicit system describing the rational parameterized curve $P(t)$ is to write it in the form $P(t) = F(A)((t^d, \dots, t))$, where F is a `TransformationFunction`, and use `FLTMD[Hd, A, d, X, Y, tol]`.

We use Example 4 to illustrate this.

Example 4 continued; define A_3 and so on.

$$\text{In[75]:= } \mathbf{A3} = \begin{pmatrix} -1 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 4 & 0 \\ 1 & 0 & 2 & 0 & 1 \end{pmatrix};$$

$$\text{In[76]:= } \mathbf{P3} = \text{TransformationFunction}[\mathbf{A3}][\{t^4, t^3, t^2, t\}]$$

$$\text{Out[76]= } \left\{ \frac{1 - t^4}{1 + 2t^2 + t^4}, \frac{4t}{1 + 2t^2 + t^4} \right\}$$

```

In[77]:= With[{d = 4},
  S4 = Table[x[i] - x[d]^(d + 1 - i), {i, d - 1}];
  H4 = Select[
    GroebnerBasis[S4, MonomialOrder -> DegreeLexicographic],
    totalDegree[#, Table[x[i], {i, d}]] == 2 &
  ]
]

```

$$\text{Out[77]= } \left\{ -x[1] + x[3]^2, -x[2] + x[3] \times x[4], -x[1] + x[2] \times x[4], \right. \\ \left. -x[3] + x[4]^2, -x[2]^2 + x[1] \times x[3], -x[2] \times x[3] + x[1] \times x[4] \right\}$$


Then we get the implicitization directly using a related function `FLTMD` (fractional linear transformation, i.e. `TransformationFunction`) that takes not points to points but equation systems to equation systems. In [6] this is simple, because all transformation matrices used are invertible. In this context the 3×5 transformation matrix is not invertible, so finding the equation system for the image of a transformation function becomes quite involved. Essentially this is the subject of all of Chapter 2 in [4]. For instance, in this case we are compacting six equations into one.

The following non-executable code and result are copied from [4, Section 3.1]. For executable code, see the `GlobalFunctionsMD.nb` notebook of [4].

$$\mathbf{G} = \text{FLTMD}[\mathbf{H4}, \mathbf{A3}, 4, \{x[1], x[2], x[3], x[4]\}, \{x, y\}, 1.*^{-12}] \\ \{1. + 2. x - 2. x^3 - 1. x^4 - 1. y^2\}$$

Once this is done we can check that this works.

```
In[78]:= piriform = 1 + 2 x - 2 x^3 - x^4 - y^2;
Show[ContourPlot[piriform == 0, {x, -2, 2}, {y, -2, 2},
  ContourStyle -> Orange],
  ParametricPlot[P3, {t, -6, 6},
  PlotStyle -> Directive[{Black, Dashed}]]]
```



■ Summary

We have shown how the Wolfram function `TransformationFunction` simplifies the study of rational parameterized curves.

■ References

- [1] J. Cardano, *The Great Art* (T. R. Whitmer, trans.), Boston: MIT Press, 1968.
- [2] J. Harris, *Algebraic Geometry, A First Course*, Springer Graduate Texts in Mathematics 133, New York: Springer, 1992.
- [3] S. Wolfram, *An Elementary Introduction to the Wolfram Language*, Champaign, IL: Wolfram Media, 2015. See also reference.wolfram.com/language.
- [4] B. H. Dayton. "Space Curve Book." <http://barryhdayton.space/SpaceCurves/spindex.html>. (Sep 4, 2020). Code in barryhdayton.space/SpaceCurves/GlobalFunctionsMD.nb.
- [5] S. Abhyankar, *Algebraic Geometry for Scientists and Engineers*, Providence, RI: American Mathematical Society, 1990.
- [6] B. H. Dayton, *A Numerical Approach to Real Algebraic Curves with the Wolfram Language*, Champaign, IL: Wolfram Media, 2018.
- [7] E. W. Weisstein. "Piriform Curve" from Wolfram MathWorld—A Wolfram Web Resource. mathworld.wolfram.com/PiriformCurve.html.

B. Dayton, "Degree versus Dimension for Rational Parametric Curves," *The Mathematica Journal*, 2020. <https://doi.org/10.3888/tmj.22-3>.

About the Author

Barry Dayton is the author of *A Numerical Approach to Real Algebraic Curves with the Wolfram Language* and is Professor Emeritus at Northeastern Illinois University in Chicago, IL. He lives in Ridgefield CT.

Barry H. Dayton

*Department of Mathematics
Northeastern Illinois University
Chicago, Illinois 60625-4699
barryhdayton.space
barry@barryhdayton.us*